Appendix

1. METAFRAME

1.1 Rationale

Based on the seven formal components a SES framework identified in Section 3 of the main paper, we have developed a compact, human-readable textual notation, called METAFRAME. The goal of this development was to provide non-technically trained SES scholars with a simply human-readable syntax that can be easily used (and has been used) in textual form via emails to exchange ideas about SES framework design decisions such as adding new concept or variables. The notation represents a SES framework in the form of nested bullet points as this has been used before by SES scholars (McGinnis and Ostrom this issue). The notation emerged through interaction with the group of authors of this paper.

At the same time, parts of METAFRAME can be directly mapped to (and hence automatically translated into) ontology languages such as OWL. In the future, once sufficient clarity on the SES framework has been gained and datasets using the framework become available, this will facilitate implementing SES-related web resources using Semantic Web technologies. We illustrate the notation with the help of the simple forestry framework introduced in Section 3 of the main paper.

1.2 Variables and data types

In METAFRAME, we use the following data types for variables:

- “String”: The set of all concatenations of symbols in the English alphabet plus punctuation symbols (i.e., the set of all natural language expressions).
- “Number”: The set of all real numbers.
- “[a,b]”: The set of all real numbers >=a and <=b.
- “Boolean”: The set of logical constants (i.e., “{true, false}”).
- “{word1, word2, word3}”: arbitrary sets of strings/words.

To express that a variable has a certain data type, we simply place the type behind a colon that follows the variable name. For variables of type “Number”, we specify the unit in which a quantity is expressed in square brackets. For example:

```
Biomass: Number [kg]
EconomicValue: Number [US$]
Precipitation: Number [mm/yr]
```

1.3 Concepts and attribution relationships

We denote attributes as bullet points written below the concept to which they are attributed to. In the simple forestry framework, for example, the variables “Biomass” and “EconomicValue” may be attributes of “Forest”. We write this as:

```
Forest
    + Biomass: Number [kg]
    + EconomicValue: Number [US$]
```

The “ + ” expresses a one-to-one attribution relationship (e.g., “A SES has one Environment”), while the “ * ” expresses a one-to-many attribution relationship left (e.g., “A SES has one or more Users”). When concepts are grouped together with further variables into higher-level concepts, an attribution hierarchy is attained. Expanding our simple forestry framework, we may take “SES” as the top concept of the hierarchy and attribute “Forest”, “User” and “Environment” (everything external to the SES) to it.

```
SES
    + Forest
        + Biomass: Number [kg]
        + EconomicValue: Number [US$]
        * Tree
            + Biomass: Number [kg]
    * User
        + Harvest: Number [kg]
        + Technology: String
    + Environment
        + Precipitation: Number [mm/yr]
```

Since all concepts at the bottom of the attribution hierarchy are variables, the whole hierarchy can be seen as a complex variable of type SES. Just as one can associate a single value to a primitive variable (e.g., harvest = 40 kg, or technology = “saw”), one can associate lists of values to the concepts, being complex variables (e.g., user = (40 kg, “saw”)).

1.4 Subsumption relationships

We express the subsumption relationship, similarly to the graphical notation, as a nested bullet-point list using the symbol “ ^ ”. For example:

```
Tree
    ^ Conifer
        ^ Pine
        ^ Larch
    ^ Broadleave
        ^ Oak
        ^ Beech
User
    ^ Lumberjack
    ^ Forester
```

1.5 Aggregation relationships

Formally, aggregation relationships are mathematical functions that map atomic variables (variables that are not disaggregated further) to aggregate ones. In METAFRAME we represent this by writing the aggregation function behind the aggregate variable. For example:

```
Forest
    + Biomass: Number [kg] = sum (Tree*.Biomass)
    * Tree
        + Biomass: Number [kg]
```

Other common aggregation functions are count, average, and standard deviation.

1.6 Outcome metrics

In our simple framework, for example, we could think of an outcome metric called "sustainability" that is computed through the biomass of the forest as well as the harvest of a user. In METAFRAME we write this as:

```
Sustainability <- (Forest.Biomass, User.Harvest)
```

1.7 Process relationships

Formally, process relationships are mathematical relations on the sets of values of the variables. From the point of view of framework development, we do not aim at representing the mathematical relation itself but only the type of relation. The former would mean that we represent a concrete model (e.g., "Biomass = 2 * Precipitation"), while the latter means that we only represent which set of variables (called domain of the mathematical relation) influences which other set of variables (called co-domain of the mathematical relation). Furthermore, we give a name to each process. The process by which the variable "Precipitation" influences "Biomass", for example, could be called "Growth". In METAFRAME we represent this as:

```
Growth: Precipitation -> Biomass
```

Or if several variables are involved:

```
Growth: (Biomass, Precipitation) -> Biomass
```

2. The refined version of the Ostrom SES framework represented in METAFRAME

The formalization of the SES framework proposed by Ostrom (2009) – with changes as proposed by Ostrom and McGinnis (this issue) – that is discussed in Section 4 of this paper can be denoted in METAFRAME as a single concept:

```
SES
    * Actor
        * SocioEconomicAttribute: String
        + ActionHistory: String
        + Location: GeographicArea
        * MentalModel: String
```

```
        + ImportanceOfResource: String
        + TechnologyAvailable: String
        ^ ActorGroup
            * Member: Actor
            + NumberOfMembers: Number = agg (Actor*)
            * SocioEconomicAttribute: String = agg (Actor*.SocioEconomicAttribute*)
            + Location: GeographicArea
            + LeadershipAndEntrepreneurship: String
            + NormsAndSocialCapital: String
            * MentalModel: String = agg (Actor*.MentalModel)
            + ImportanceOfResource: String = agg (Actor*.ImportanceOfResource)
            + TechnologyAvailable: String
            ^ Organization
                ^ NonGovernmentalOrganization
                ^ GovernmentalOrganization
    + GovernanceSystem
        + NetworkStructure: String
        * PropertyRightsSystem: String
        * RulesInUse
            ^ ConstitutionalRule: ADICO
            ^ CollectiveChoiceRule: ADICO
            ^ OperationalRule: ADICO
                ^ MonitoringAndSanctioningRules: ADICO
    * ResourceSystem
        + Sector: {'water', 'forestry', 'pasture', 'fishery'}
        + ClarityOfSystemBoundary: [0,1]
        + Location: GeographicArea
        * HumanConstructedFacility
        + Productivity: [0,1]
        * EquilibriumProperty: String
        + PredictabilityOfSystemDynamics: [0,1]
        + StorageCharacteristic: String
        * PopulationOfResourceUnits
            + Mobility: String
            + GrowthOrReplacementRate: Number = agg (ResourceUnit*.GrowthOrReplacementRate)
            + InteractionAmongUnits: Boolean
            + EconomicValue: Number = agg (ResourceUnit*.EconomicValue)
            + NumberOfUnits: Number = agg (ResourceUnit*)
            + DistinctiveCharacteristics: Boolean
            + SpatialAndTemporalDistribution: String = agg (ResourceUnit*.Location)
            * ResourceUnit
                + Mobility: String
                + GrowthOrReplacementRate: Number [kg/yr]
                + EconomicValue: Number [USD/kg]
                + DistinctiveCharacteristics: Boolean
                + Location: GeographicPoint
    + Environment
        * SocialEconomicPoliticalSettings
        * RelatedEcosytem
```

Notes on design decisions:

This formalization is by no means the definitive version of the formal SES framework. As explained in the main text of this article, framework development is a dynamic process. At any time, SES researchers may introduce new variables as part of their inquiry of a particular SES, or comparison across multiple SESs. Accommodating these new variables within the framework may lead to the definition of new concepts and/or modification of existing concepts.

ResourceSystem:

- the value of sector (RS1) is one of a limited set of string values;
- clarity of system boundaries (RS2) and predictability of system dynamics (RS7) can be expressed on a scale from 0 (low) to 1 (high);
- equilibrium properties (RS6) and storage characteristics (RS8) are (for now) described in natural language; and
- By representing location (RS9) as a geographical area, we also include the variable RS3 (Size of resource system)

ResourceUnit:

- The name of variable RU3 (Interaction among resource units) suggests a process, and these we would formalize as process relationships.
- We decided to define "Mobility" to be of type "String", because defining it by means of an aggregation relation with Location (e.g., movement rate in m/day) may not capture the information that resource units are territorial, or that their movement is confined. The scale units for growth rate and economic value may also be subject to debate.
- "PopulationOfResourceUnits.GrowthOrReplacementRate" is not necessarily an aggregation of "ResourceUnits.GrowthOrReplacementRate", because in fisheries, for example, the growth rate of a population relates to an increase in numbers or biomass while the growth rate of an individual is often measured as an increase in length.

Actors

- We struggled with the alternative to see "Organization" as a direct sub-concept of "Actor" and not of "ActorGroup". While clearly an organization "is-a" special kind of actor group, it is not clear that one would want to describe the internals of an organization in terms of all ActorGroup variables such as social capital and leadership. The point is that in any analysis of SES one needs to decide which social entities to resolve and describe internally and which to treat as unresolved holistic collective actors and one could regard organizations as the latter type.
- We give "location" (A4) the data type "GeographicArea" because this offers more flexibility than single point locations.

Formalization of institutions in ADICO

```
ADICO
    ^ SharedStrategy
        + ActorGroup
        + Condition: String
        + Aim: String
        ^ Norm
            + Deontic: {'must', 'may', 'must not'}
            ^ Rule
                + OrElse: String
                + Level: {'operational', 'collective-choice', 'constitutional'}
                + Type: {'position', 'boundary', 'choice', 'aggregation', 'information', 'payoff', 'scope'}
```

This concept hierarchy reflects that rules are the most specific form of institution. A rule specifies for a particular group of actors under which condition it *must* (obligation), *may* (permission or right), or *must not* (prohibition) aim for a particular goal or action, *or else* risk certain sanctions. A norm is a rule without sanctions; a shared strategy has no deontic and hence merely states under which conditions a group is willing to take collective action. Note that we have added the attribute `Level` to allow distinguishing between the three types of rules.

3. A framework of a recreational fishery case represented in METAFRAME

```
RecreationalFisheriesSES
    * AnglingClub
        + NumberOfMembers: Number = agg(Member*)
        + TotalCatch: Number
        * StockingRule
        * Member
            + Attitude: String
            * Position: {angler, memberOfBoard}
            ^ Angler
                + Satisfaction: Number
                + Catch: Number
                + Effort: Number
    * WaterBody
        + HabitatConditions: String
        * FishPopulation
            + GrowthOrReplacementRate: Number
            + NumberOfFish: Integer = agg(Fish*)
            * Fish
                + Origin: {‘wild’, ‘hatchery’}
    + Environment
        + Climate
        + HatcheryFishMarket
```

4. The refined Ostrom framework merged with the framework of the recreational fishery case.

This appendix shows how the Ostrom framework (Section 2) is extended with the framework of the recreational fishery case (Section 3). The black concepts/variables are those of the general SES framework that have not been directly used in the recreational fishery case, the blue concepts are those that have and the red concepts are those that have newly been added.

```
SES
    * Actor
        * SocioEconomicAttribute: String
        + ActionHistory: String
        * Position: {user, provider, …}
        * MentalModel: String
        + ImportanceOfResource: String
        + Location: GeographicArea
        + TechnologyAvailable: String
        ^ Angler
            + Satisfaction: Number
            + Catch: Number
            + Effort: Number
        ^ ActorGroup
            * Member: Actor
            + NumberOfMembers: Number = agg (Actor*)
            * SocioEconomicAttribute: String = agg (Actor*.SocioEconomicAttribute*)
            + Location: GeographicArea
            + LeadershipAndEntrepreneurship: String
            + NormsAndSocialCapital: String
            * MentalModel: String = agg (Actor*.MentalModel)
            + ImportanceOfResource: String = agg (Actor*.ImportanceOfResource)
            + TechnologyAvailable: String
            ^ Organization
                ^ GovernmentalOrganization
                ^ NonGovernmentalOrganization
                    ^ AnglingClub
                        + TotalCatch: Number
    + GovernanceSystem
        + NetworkStructure: String
        * PropertyRightsSystem: String
        * RulesInUse
            ^ ConstitutionalRule: ADICO
            ^ CollectiveChoiceRule: ADICO
            ^ OperationalRule: ADICO
                ^ MonitoringAndSanctioningRules: ADICO
                ^ StockingRule
    * ResourceSystem
        + Sector: {'water','forest','pasture','fishery','recreational fishery'}
        + ClarityOfSystemBoundary: [0,1]
        + GeographicArea
        * HumanConstructedFacility
        + Productivity: [0,1]
        * EquilibriumProperty: String
        + PredictabilityOfSystemDynamics: [0,1]
        * StorageCharacteristic: String
        ^ WaterBody
            + HabitatConditions: String
        * PopulationOfResourceUnits
            + Mobility: String
            + GrowthOrReplacementRate: Number = agg (ResourceUnit*.GrowthOrReplacementRate)
            + InteractionAmongUnits: Boolean
            + EconomicValue: Number = agg (ResourceUnit*.EconomicValue)
            + NumberOfUnits: Number = agg (ResourceUnit*)
```

    + DistinctiveCharacteristics: Boolean
    + SpatialAndTemporalDistribution: String = agg (ResourceUnit*.Location)
    * ResourceUnit
      + Mobility: String
      + GrowthOrReplacementRate: Number
      + EconomicValue: Number
      + DistinctiveCharacteristics: Boolean
      + Location: GeographicPoint
      ^ Fish
        + Origin: {‘wild’, ‘hatchery’}
+ Environment
  * SocialEconomicPoliticalSettings
    ^ HatcheryFishMarket
  * RelatedEcosytem
    ^ Climate